

Méthode des éléments finis: TP2

Ibrahim ALAME

11/03/2024

Prise en Main de Cast3M

Présentation de Cast3M


Castem2000 est un logiciel de calcul de structures par la méthode des éléments finis et plus généralement de résolution d'équations aux dérivées partielles par la méthode des éléments finis. Il a été développé par le Commissariat à l'Énergie Atomique (CEA). Depuis l'été 1999, *Castem2000* est gratuit pour l'enseignement et la recherche, de plus le code source est ouvert aux développeurs. Il est devenu *Cast3M* en 2001. Pour débiter avec *Cast3M*, le point d'entrée est le site du CEA : <http://www-cast3m.cea.fr/> A mettre dans vos favoris, l'utilisation de ce site est indispensable pour travailler avec *CAST3M*.

Installation

La première étape de ce TP consiste à installer le code *Cast3M* à partir du site du CEA sur votre ordinateur (vous y avez droit en tant qu'étudiant). Nous ne détaillons pas ici les étapes de ces installations, il suffit de suivre les instructions qui vous seront données ici.

On pourra lancer *Cast3M* à partir d'une console à l'aide de la commande *castem22* mais je vous conseille plutôt d'utiliser l'éditeur *Visual Studio* que vous pouvez installer et paramétrer de la façon suivante :

1. Installez *Visual Studio Code* pour votre plateforme depuis ce lien : <https://code.visualstudio.com/download>.
2. Une fois dans l'IDE, cliquez sur l'icône ci-dessous (1).



3. Tapez *Gibiane* dans la barre de recherche (2) et sélectionnez le premier résultat (3).
4. Cliquez sur le bouton *Installer* (4).
5. Vous pouvez maintenant ouvrir votre fichier *.dgibi*.

La configuration est à faire uniquement si le bouton d'exécution des scripts ne fonctionne pas ! Si c'est le cas cliquez sur la roue dentée (1) comme indiqué sur la copie d'écran à la page suivante, puis sur **settings** (2), puis dans la barre de recherche (3) tapez **Gibiane**. Dans le champs **Gibiane-vscode** :Commande (4), rentrez le nom de la commande avec laquelle vous exécutez votre script habituellement. Par exemple, si vous utilisez *castem22 /home/me/script.dgibi*, rentrez **castem22** uniquement.

Cast3M possède une documentation en ligne. On peut (et on doit) la consulter pour avoir la syntaxe d'une commande donnée. Cette documentation est au format html, on peut donc la visualiser à l'aide d'un navigateur. Une copie de cette documentation peut être consultée à la page <http://www-cast3m.cea.fr/index.php?page=notices> ou bien http://www-cast3m.cea.fr/index.php?page=notices_classees

Le langage de Cast3M : Gibiane

Gibiane est le langage interprété qui permet de communiquer avec le programme. Ainsi, le principe est d'écrire un programme en langage GIBIANE à l'aide d'un éditeur de texte (n'importe lequel). Puis de lancer l'application *castem* sur le fichier créé. Il est recommandé d'utiliser le suffixe *.dgibi*.

```

54
55 * dessin de la ligne constituant le contour
56 TITRE TITRE1 'Contour';
57 TRAC CONT1;
58
59 * Maillage
60
61 MAIL1 = SURF CONT1 COUL 'ROUGE' ;
62
63 MAIL2 = DALL AD AB (BC et EC) DE COUL 'VERT';
64 * dessin du maillage
65 TITRE TITRE1 'Maillage et Contour';
66 TRAC MAIL1;
67
68 * ++++++
69 * TRAITEMENT DU PROBLEME
70 * ++++++
71 * Definition du modele du calcul
72
73 MODL1 = MODE MAIL2 'MECANIQUE' 'ELASTIQUE' 'ISOTROPE' ;
74
75 * Definition du materiau utilise
76 * Caracteristiques du materiau
77 YOUN1 = 200e3;

```

La syntaxe est basée sur l'utilisation de directives, d'opérateurs et de procédures qui s'appliquent à des opérandes. Dans le premier cas la syntaxe est :

DIRECTIVE OPERANDES.

Par exemple dans

TRACE MAILLAGE ;

TRACE est la directive de traçage et **MAILLAGE** est l'opérande que l'on veut visualiser. Dans un second cas, la syntaxe est :

RESULTATS = OPERATEUR OPERANDES;

Par exemple dans

LIGNE = DROITE P1 P2 ;

LIGNE sera l'objet construit en reliant **P1** à **P2** par une droite. **DROITE** est l'opérateur qui s'applique sur les opérandes **P1** et **P2** et le résultat s'appelle **LIGNE**. La procédure peut utiliser, suivant sa définition, l'une ou l'autre des syntaxes. Il convient de compléter ce paragraphe en précisant quelques règles syntaxiques de **GIBIANE** :

- Le point-virgule termine une instruction.
- Gibiane ne fait pas a priori de distinction entre Majuscule ou en minuscule.
- Une ligne peut contenir plusieurs instructions (séparées par des points-virgules).
- Les lignes de commentaire sont précisées par un astérisque dans la première colonne.
- Les opérateurs et les directives sont définis par leurs 4 premiers caractères mais on peut en donner plus s'il n'y a pas de confusions possibles (ex : **TRAC**, **TRACE**, **TRACER**). Il est conseillé d'utiliser un mot lisible : **TRACER** plutôt que **TRAC** et **DROITE** plutôt que **DROI**.
- L'instruction est interprétée de gauche à droite, et les opérateurs sont exécutés dès qu'ils sont lus. Ainsi $1+2*3=9$ (!) . Pour retrouver l'ordre de priorité mathématique, il convient d'ajouter des parenthèses : $1+(2*3)=7$.
- La longueur du nom attribué à un objet ne doit pas dépasser 8 caractères. Il est conseillé d'éviter d'attribuer à un objet le nom d'un opérateur existant, car ce dernier serait alors écrasé. Pour cela on peut conseiller de ne pas donner des noms de 4 caractères et de mettre un nombre en fin de nom (il n'y a qu'un opérateur ayant un nombre en fin de nom c'est **CER3**) ou de protéger les opérateurs et directives en utilisant les côtes (ex : '**DROI**').

Système d'unités

Castem ne dispose d'aucun système particulier d'unité de mesure. Seule la mesure des angles doit être dans tous les cas exprimée en degrés pour la géométrie alors que les résultats obtenus sont en radians.

Éléments finis par Cast3M

Tout problème d'éléments finis peut être construit de la manière suivante :

1. Description de la géométrie, maillage. Choix du support géométrique.
2. Choix du type d'éléments finis et du modèle de comportement.
3. Donnée des caractéristiques du matériau et des caractéristiques géométriques supplémentaires.
4. Donnée des conditions aux limites.

- Donnée du chargement.
- Résolution du système.
- Post-traitement des résultats.

Génération de Maillage

L'objet du maillage est de discrétiser géométriquement le domaine d'analyse de manière à pouvoir ultérieurement associer une formulation éléments finis au support géométrique. Concrètement cette discrétisation s'effectue par la création d'objets de type maillage (points, lignes, surfaces, volumes) à l'aide des opérateurs géométriques. La technique à suivre est presque toujours la même :

- construction des points
- construction des lignes à partir des points
- construction des surfaces à partir des lignes
- construction des volumes à partir des surfaces

Maillage Simple

Dans un premier exemple, nous allons mailler un cube de côté 10 m (l'unité en mètre est ici indiquée pour fixer les idées, rappelons qu'il n'y a pas de système d'unité sur Castem).

```

1 *
2 * DEBUT DU FICHIER MAILLAGE-CUBE.DGIBI
3 *
4 TITR 'PREMIER MAILLAGE' ;
5 OPTI DIME 3 ELEM CUB8 ;
6 *
7 * POINTS
8 *
9 P1 = 0. 0. 0. ;
10 P2 = 10. 0. 0. ;
11 P3 = 10. 10. 0. ;
12 P4 = 0. 10. 0. ;
13 *
14 * LIGNES
15 *
16 P1P2 = P1 DROITE 10 P2 ;
17 P2P3 = P2 DROITE 10 P3 ;
18 P3P4 = P3 DROITE 10 P4 ;
19 P4P1 = P4 DROITE 10 P1 ;
20 *
21 * SURFACE
22 *
23 SURF1 = DALLER P1P2 P2P3 P3P4 P4P1 ;
24 *
25 * VOLUME
26 *
27 VOL1 = SURF1 VOLUME 10 TRANSLATION (0. 0. 10.) ;
28 *
29 * TRACAGE
30 *
31 TRACER SURF1 QUALIFICATION ;
32 TRACER CACHER VOL1 ;
33 *
34 * FIN DU FICHIER
35 *
36 FIN ;

```

- La directive OPTI(ON) permet de déclarer les principaux paramètres du programme, par exemple la dimension de l'espace, le type d'éléments géométriques utilisés....
- Le point est construit en associant à son nom, ses coordonnées.
- L'opérateur DROI(TE) permet de construire une ligne droite à partir de ses deux points extrêmes. Ces lignes créées sont orientées et automatiquement subdivisées en un certain nombre de segments que l'on pourra spécifier. Ici on a spécifié 10 segments entre deux points.
- L'opérateur DALL(ER) permet de construire une surface délimitée par 4 côtés ayant deux à deux le même nombre de points et formant une ligne fermée.
- L'opérateur VOLU(ME) permet de construire des volumes par translation de surface suivant un vecteur avec l'option TRAN(LATION).
- La directive TRAC(ER) trace un objet de type maillage (ici la surface de base). L'option QUAL(IFICATION) permet d'afficher les noms des objets visualisés à l'écran.

Premier calcul mécanique linéaire

3.1 Objectif du calcul Le but de ce calcul est de voir sur un exemple très simple l'enchaînement des étapes nécessaires à un calcul par éléments finis par CASTEM. Pour cela on veut calculer la déformée d'une poutre encastrée à une extrémité et subissant une force fléchissant à l'autre extrémité. Les données du problème sont :

longueur : $L = 1\text{m}$, section : $S = 1.392 \times 10^{-4}\text{m}^2$, masse volumique : $\rho = 7800\text{kg/m}^3$,
inertie suivant y et z : $I_{yy} = I_{zz} = 2.673 \times 10^{-4}\text{m}^4$, inertie de torsion : $I_T = 1.0 \times 10^{10}$,
force appliquée : $F = 10\text{N}$, module d'Young : $E = 210000\text{MPa}$,

```

1 TITR 'PREMIER CALCUL MECANIQUE' ;
2 OPTI DIME 3 ELEM SEG2 ;
3 *
4 * GEOMETRIE
5 *
6 P1 = 0. 0. 0. ;
7 P2 = 1. 0. 0. ;
8 POUTRE1 = P1 DROI 10 P2 ;
9 *
10 * CHOIX DU COMPORTEMENT DU MATERIAU ET DU MODELE EF
11 *
12 M1 = MODE POUTRE1 MECANIQUE ELASTIQUE ISOTROPE POUT ;
13 *
14 * DONNEES MATERIAU ET ELEMENT
15 *
16 CARMAT1 = MATE M1 YOUNG 2.1E11 NU 0.3 RHO 7.8E3 ;
17 CAREF1=CARA M1 SECT 1.392E-4 INRY 2.673E-10 INRZ 2.673E-10 TORS 1.E10 ;
18 CARTOT1 = CARMAT1 ET CAREF1 ;
19 *
20 * CONDITIONS AUX LIMITES
21 *
22 CONDI1 = BLOQ DEPL ROTA P1 ;
23 *
24 * CHARGEMENT
25 *
26 FOR1 = FORC (0. 0. -10.) P2 ;
27 *
28 * RESOLUTION
29 *
30 RIGI1 = RIGI M1 CARTOT1 ;
31 RIGICL1 = RIGI1 ET CONDI1 ;
32 DEPL1 = RESO RIGICL1 FOR1 ;
33 *
34 * POST-TRAITEMENT
35 *
36 F1 = EXTR DEPL1 UZ P2 ;
37 F1 = 1.E3*F1 ;
38 MESS 'FLECHE EN P2' F1 'mm' ;
39 OEIL1 = 0. -1000. 0. ;
40 DEF0 = DEFO POUTRE1 DEPL1 0. BLAN ;
41 DEF1 = DEFO POUTRE1 DEPL1 ROUG ;
42 TRAC OEIL1 (DEF0 ET DEF1) ;
43 *
44 * FIN DU FICHIER
45 *
46 FIN ;

```

Le résultat recherché est la flèche à l'extrémité.

Analyse de la solution

Début de l'étude :

```
TITR 'PREMIER CALCUL MÉCANIQUE' ; *Donne un nom à l'étude.  
OPTI DIME 3 ELEM SEG2 ;
```

Le problème mécanique nous amène à choisir des éléments finis de type poutre. La documentation nous informe que les éléments géométriques correspondant à l'élément POUT sont des SEG2 et que la dimension requise pour ce type d'étude est 3.

Définition de la géométrie :

```
P1 = 0. 0. 0. ;  
P2 = 1. 0. 0. ;  
POUTRE1 = P1 DROI 10 P2 ;
```

La géométrie est décrite en plaçant les points extrémités P1 et P2, et en traçant une ligne droite entre P1 et P2 contenant 10 segments.

Choix du modèle de comportement et du type d'Eléments Finis :

```
M1=MODE POUTRE1 MECANIQUE ELASTIQUE ISOTROPE POUT ;
```

L'opérateur MODE(LE) sert à définir un type de comportement et une formulation E.F. qui seront affectés à un objet de type maillage. L'objet créé (ici M1) est de type MMODEL. On a affecté sur la géométrie POUTRE1 des éléments finis de type POUT et une loi de comportement mécanique élastique isotrope.

Entrée des caractéristiques des matériaux et de la MEF

```
CARMAT1=MATE M1 YOUN 2.E11 NU 0.3 RHO 7.8E3 ;
```

MATE(RIAU) sert à définir les propriétés physiques du matériau (Module d'Young, coefficient de Poisson...) pour un modèle donné. L'objet créé est de type champ par élément (MCHAML) à plusieurs composantes : YOUN, NU, RHO, ALPH,...

```
CAREF1=CARA M1 SECT..... TORS 1.E10 ;
```

Certains éléments nécessitent la donnée de caractéristiques supplémentaires qui ne peuvent se déduire de la géométrie. Dans le cas des poutres (POUT) il faut préciser SECT, INRY, INRZ, TORS. Dans le cas des barres (BARR) il suffit de préciser la section.

```
CARTOT1=CARMAT1 ET CAREF1 ;
```

On prend en compte dans CARTOT1 l'ensemble des caractéristiques définies. On peut préciser ici que tout cela aurait pu s'écrire en une seule ligne :

```
CARTOT1=MATE M1 YOUN.....TORS 1.E10 ;
```

Cependant, il nous a semblé pédagogique de séparer les caractéristiques suivant leurs origines. D'autre part, il ne faut pas oublier qu'une ligne d'instructions ne doit comprendre que 72 caractères au maximum.

Conditions aux limites

```
COND1=BLOQ DEPL ROTA P1 ;
```

Les conditions aux limites sont traitées dans CAST3M comme une rigidité à adjoindre à la rigidité du système libre grâce à l'opérateur BLOQ(UEP)

Conditions de chargement :

FOR1=FORC (0. 0. -10.) P2 ;

Définir le chargement revient à définir un champ par point correspondant au vecteur du second membre de $[K]\{u\} = \{f\}$. Ici on applique une force de -10N suivant l'axe z au point P2. Ceci s'effectue avec l'opérateur FORC(E). On aurait pu également écrire

FOR1=FORC FZ -10. P2 ;

La syntaxe de l'opérateur MOME(NT) est du même type. L'opérateur pression, sera quant à lui étudié ultérieurement.

Résolution :

L'ensemble des données étant défini, on peut constituer le système $[K]\{u\} = \{f\}$ et le résoudre.

RIGI1=RIGI M1 CARTOT1 ;

L'opérateur RIGI(DITE) permet de construire la matrice de rigidité à partir du modèle et des caractéristiques relatives au modèle.

RIGICL1=RIGI1 ET COND1 ;

Comme nous l'avons dit dans le paragraphe conditions aux limites, il convient de prendre en compte la matrice des blocages au sein de la matrice rigidité.

DEPL1=RESO RIGICL1 FOR1 ;

L'opérateur RESO(UT) résout le système $[K]\{u\} = \{f\}$. Ces déplacements, solutions du problème, sont stockés dans DEPL1.

Post traitement :

F1=EXTR DEPL1 UZ P2 ;

L'opérateur EXTR(AIRE) permet d'extraire une composante d'un ensemble de valeurs. Ici on recherche le déplacement en UZ du point P2 au sein du vecteur solution en déplacement DEPL1.

F1 = 1.E3 * F1 ;

La résolution par RESO donne des résultats en mètres avec notre choix d'unité. Pour l'avoir en mm on le multiplie donc par 1000.

MESS 'FLECHE EN P2' F1 'mm' ;

La directive MESS(AGE) permet d'afficher un message sur l'unité de sortie.

OEIL1 = 0. -1000. 0. ;

On définit ici le point à partir duquel sera vue la structure lors des visualisations. On peut ainsi définir de multiples points de vue.

DEFO = DEFO POUTRE1 DEPL1 0. BLAN;

DEF1 = DEFO POUTRE1 DEPL1 ROUG ;

TRAC OEIL1 (DEFO ET DEF1) ;

L'opérateur DEFO(RME) construit la déformée d'une structure à partir de la géométrie initiale et d'un champs de déplacement. On peut également préciser un certain nombre d'options comme la couleur (ici BLAN(C) et ROUG(E)), ou le facteur d'amplification pour rendre les phénomènes plus visibles. Ici on utilise un facteur multiplicatif de 0 sur DEFO. Ceci permet de visualiser la structure non déformée en même temps que la déformée finale DEF1. Cet artifice est nécessaire car la directive TRAC(ER) ne peut être appliquée qu'à des objets de même type. On ne pourrait donc pas avoir :

TRAC OEIL1 (POUTRE1 ET DEF1)

car l'un est de type maillage et l'autre déformée. On remarque également l'utilisation du point OEIL1 pour préciser le point de vue selon lequel on doit effectuer le traçage des déformés.

fin du programme

FIN ;

La directive FIN permet de quitter CASTEM.

Poutre en dim 2

```
*****
* Poutre 1D élastique
*****
*
* fichier poutre1D.dgibi
*
TITR 'Poutre 1D élastique' ;
OPTI DIME 2 ELEM SEG2 ;
*
* Données du problème (Unités : m, Pa)
*
F = -100 ;
L = 1.00 ;
h1 = 0.02 ;
b1 = 0.01 ;
E = 2E11 ;
nu1 = 0.3 ;
Izz = (h1**3)*b1/12 ;
S1=b1 * h1;
*
* Solution analytique RdM
*
fana0 =F*(L**3)/(3*E*Izz);
*
* Géométrie
*
P1=0. 0. ;
P2 = L 0. ;
NBelements = 10 ;
Poutre1 = P1 DROI NBelements P2 ;
TRAC Poutre1;
*
* Modèle (Comportement et modélisation EF)
*
mo1=MODE Poutre1 MECANIQUE ELASTIQUE ISOTROPE POUT ;
*
* Données matériau et élément
*
ma1=MATE mo1 YOUN E SECT S1 NU nu1 INRZ IZZ;
*
* Conditions aux limites
*
Condi =BLOQ UX UY RZ P1 ;
*
* Chargement
*
Force1=FORC (0. F) P2 ;
*
* Résolution
*
Rigi1=RIGI mo1 ma1 ;
RigiCL1=Rigi1 ET Condi ;
Dep11=RESO RigiCL1 Force1 ;
*
* Poste Traitement
*
Fleche1=EXTR Dep11 UY P2 ;
*
MESS 'Flèche CASTEM : ' Fleche1 'm' ;
MESS 'Flèche analytique : ' fana0 'm' ;
*
Def0=DEFO Poutre1 Dep11 0. BLAN ;
Def1=DEFO Poutre1 Dep11 ROUG ;
TRAC (Def0 ET Def1) ;
*
* Fin du fichier
*
FIN ;
```

Exercices

1. Maillage d'un cube

A l'aide d'un éditeur de texte, modifier le fichier maillage-cube.dgibi de façon à construire un parallélépipède de base carré de côté 5m et de hauteur 20m. Essayer d'avoir un maillage régulier (i.e. vraiment cubique!). Sauvegarder le fichier sous un autre nom.

On veut remplacer la commande DALL(ER) par SURF(ACE). Commencez par consulter la documentation concernant ces commandes.

1. Définir à l'aide de la commande ET un contour fermé carré.

2. Puis définir la surface carré à l'aide de la commande SURF(ACE) avec l'option 'PLAN(E)'.
3. Essayer les autres options juste pour voir.

2. Maillage d'un cylindre

A l'aide des instructions CERC(LE), SURF(ACE) et VOLU(ME) nous allons construire un cylindre de base circulaire de centre (0, 0, 0) de rayon 2m et hauteur 10m. Pour obtenir la syntaxe de ces commandes, consulter la documentation en ligne.

1. Pour commencer, construire 4 quarts de cercle avec la commande CERC(LE).
2. Réunir les quatres morceaux précédent avec la commande ET.
3. Utiliser la commande SURF(ACE) avec l'option 'PLAN(E)' pour construire un disque.
4. Achever la construction du cylindre à l'aide de la commande VOLU(ME).
5. Construire un cylindre creux de hauteur 10m et dont la base est une couronne délimité par deux cercles de centre (0, 0, 0) et de rayon respectifs 2m et 2.5m. Indication : tracer les deux cercles comme précédemment, les réunir à l'aide de la commande ET puis définir la couronne à l'aide de la commande SURF(ACE). Essayer d'obtenir un maillage régulier.

3. Maillage d'une plaque trouée

1. En dimension 2, créer le maillage d'une plaque carré de coté 5m troué d'un disque de rayon 1m. Indication construire le contour du carré puis du cercle. Les réunir, puis utiliser ensuite la commande SURF(ACE).
2. Même chose en dimension 3.

3. Treillis

Exécuter puis modifier le fichier `barresArticulees.dgibi` disponible sur e-campus afin d'étudier la structure des barres articulées de TP1. Comparer avec le résultat de TP1.

4. Modélisation d'un pont

On considère un pont plan de dimensions $L = 20$ mm, $H = 8$ mm et d'épaisseur $e = 5$ mm. Son ouverture est en demi cercle central de rayon $R = 5$ m. Ses caractéristiques élastiques sont : $E = 200$ GP a, $\nu = 0.3$. La charge appliquée est linéique de densité $q = e \times \sigma$ avec $\sigma = 500$ MP.

1. On suppose les contraintes et les déformations planes et on choisit d'étudier le pont dans son plan de symétrie verticale et on choisit comme élément fini de référence le triangle de type (1) TRI3 :

```
OPTION DIMENSION 2 ELEMENT TRI3 MODE PLAN CONTRAINTES;
```

2. Réaliser un maillage de la plaque avec une densité $\frac{R\pi}{15}$ (équivalent à 15 éléments autour de la partie circulaire) :

```
DENSITE R*PI/15;
```

3. Tracer le maillage avec l'option de calification :

```
CONT1=D1 ET D2 ET D3 ET D4 ET D5 ET C1;
SURF1=SURFACE CONT1 PLANE;
TRACER SURF1 QUAL;
```

4. Définir le modèle mécanique ainsi que les caractéristiques physiques :

```
MOD1=MODELE SURF1 MECANIQUE ELASTIQUE PLASTIQUE PARFAIT;
MAT1=MATE MOD1 YOUN 2.1E11 NU 0.28 DIM3 EP1 SIGY SIGE;
```

5. Appliquer les conditions aux limites en bloquant les déplacements en x et y le long de $D1$ et $D5$ et en imposant une contrainte σe le long de $D3$:

```
CL1=BLOQ D1 ..... ; CL2=BLOQ .... .....;
F1=PRESSION MASSIF MOD1 D3 (SIGE*EP1);
```

6. Définir la rigidité matérielle en appliquant l'opérateur RIGI au modèle mécanique MOD1 en tenant compte des caractéristiques physiques MAT1 :

```
RIG1 = RIGI .... ..;
```

7. Assembler avec les raideurs dues aux conditions aux limites (CL1 et CL2) avec la rigidité matérielle RIG1 pour obtenir la rigidité totale RITOT :

RITOT = ET ET;

8. Résoudre à l'aide de l'opérateur RESOU, le système linéaire $KU = F$ où K est la matrice de rigidité totale notée ici RITOT, U est le champ de déplacement inconnu noté DEP1 et F le second membre noté ici F1

DEP1=RESOU ;

9. Tracer enfin, l'état de contrainte sur la déformée en faisant :

DEF1=DEFO DEP1 SURF1;

SIG1=SIGMA DEP1 MOD1 MAT1;

TRACER SIG1 MOD1 DEF1;

10. Refaire la même étude en remplaçant le demi cercle par un rectangle de largeur $2R$ et de hauteur R .

